```
/****SAS Codes Solution for Q1*******/
legend1 frame across=1 label=("Sales Channels");
AXIS1 order=(1 to 13 by 1) minor=NONE label=("Month in 2007");
AXIS2 order=(200 to 600 by 30) minor=NONE label=("Sales Amount") major=(h=1.1);

symbol1 c=RED v=dot i=spline;
symbol2 c=BLACK v=dot i=spline;
symbol3 c=BLUE v=dot i=spline;

proc gplot data=Groceries;
  plot sales_amt_1*month
       sales_amt_2*month
       sales_amt_3*month /
       overlay ctext=BLACK LEGEND=LEGEND1
  HAXIS=AXIS1 VAXIS=AXIS2;
  where year=2007;
  title 'Sales Trend in 2007 with Three Channels';
run;


/****SAS Codes Solution for Q2*******/
legend1 frame across=1 label=("Year");
AXIS1 order=(1 to 13 by 1) minor=NONE label=("Month in 2007");
AXIS2 order=(220 to 280 by 10) minor=NONE label=("Sales Amount") major=(h=1.1);

symbol1 c=RED v=dot i=spline;
symbol2 c=BLACK v=dot i=spline;

proc gplot data=Groceries;
  plot sales_amt_1*month=year
       / ctext=BLACK LEGEND=LEGEND1
         HAXIS=AXIS1 VAXIS=AXIS2;
  where year in (2007,2008);
  title 'Sales Trend in 2007 and 2008 for Channel 1';
run;
```

```
/****SAS Codes Solution for Q3*******/
pattern1 value=SOLID color=BLUE;
pattern2 value=SOLID color=RED;
pattern3 value=SOLID color=YELLOW;
AXIS1  label=("Month");
AXIS2  label=("Sales $ Amount 2");

proc gchart data=Groceries;
  VBAR3D month
  /CTEXT=BLUE sumvar=sales_amt_2 midpoints=1 to 12 by 1
       subgroup=year  MAXIS=AXIS1 RAXIS=AXIS2 type=mean;
  where year in (2007,2008,2009);
  Title 'YOY AVG Sales for Channel 2';
run;


/****SAS Codes Solution for Q4*******/
pattern1 value=PSOLID color=RED;
pattern2 value=PSOLID color=BLACK;
pattern3 value=PSOLID color=P;
legend1 label='year' position=(left middle) value=(color=green);

proc gchart data=Groceries;
  format sales_amt_1 dollar11. year 8.;
  PIE year /CTEXT=BLACK NOHEADING sumvar=sales_amt_1 type=mean
  PLABEL=(COLOR=BLUE HEIGHT=2 FONT="Arial")
  value=inside legend=legend1 percent=arrow;
  where year in (2007,2008,2009);
  Title 'AVG Sales for Channel 1';
run;
quit;


/****SAS Codes Solution for Q5*******/
PROC format;
  VALUE agef
        10 -< 30 = '20-30'
        30 -< 50 = '30-50'
        50 -< 70 = '50-70'
        70 - HIGH = '70+'
        OTHER ='N/A';
Run;

data risk;
  set risk_data;
```

```
  utilization=utilization/100;
  agegrp=put(age,agef.);
  format utilization percent11.1;
run;

proc sort data=risk;
  by age;
run;

PROC TEMPLATE;
  DEFINE STATGRAPH bartemp;
  BeginGraph;
  EntryTitle "Age and Bad Rate";
  layout overlay /yaxisopts=(label="Bad Rate");

  barchart x=agegrp y=high_risk_ind / stat=mean;

  endlayout;
 EndGraph;
 END;
RUN;

proc sgrender data=Risk TEMPLATE=bartemp;
  format high_risk_ind percent11.1;
run;
quit;


/****SAS Codes Solution for Q6*******/
proc sort data=risk;
  by age;
run;
proc means data=Risk;
  var utilization high_risk_ind age income;
  by agegrp;
  output out=bucket (drop=_freq_ _type_) mean=utilization high_risk_ind age income;
run;

PROC TEMPLATE;
  DEFINE STATGRAPH bartemp;
  BeginGraph;
  EntryTitle "Age, utilization and Bad Rate";
  layout overlay /yaxisopts=(label="AVG Utilization")
                 xaxisopts=(label="Age Group");

   barchart x=agegrp y=utilization  / stat=mean;
```

```
   seriesplot x=agegrp y=high_risk_ind /curvelabel="Bad Rate";
  endlayout;
 EndGraph;
 END;
RUN;

proc sgrender data=bucket TEMPLATE=bartemp;
  format high_risk_ind Utilization percent11.1;
run;
quit;


/****SAS Codes Solution for Q7*******/
proc rank data=risk_data groups=10 out=Risk_ut;
  var utilization;
  ranks ugrp;
run;
proc sort data=Risk_ut;
  by ugrp;
run;
proc means data=Risk_ut;
  var utilization high_risk_ind age income;
  by ugrp;
  output out=bucket (drop=_freq_ _type_) mean=utilization high_risk_ind age income;
run;

proc template;
 define statgraph lat;
  begingraph;
   entrytitle "profiling on income, age and risk level";
   layout lattice;

     Layout overlay/yaxisopts=(label="Bad Rate");
        barchart x=ugrp y=high_risk_ind ;
     endlayout;

     barchart x=ugrp y=income;
     barchart x=ugrp y=age;
   endlayout;
  endgraph;
 end;
run;

proc sgrender data=bucket template=lat;
run;
```

```
/****SAS Codes Solution for Q8*******/
proc template;
  define statgraph dynagr;
    begingraph;
            dynamic var1 var2 xlabel ylabel plottitle;
            mvar SYSDATE9 statistics axislabelcolor barfillcolor;
            nmvar axislabelsize ;
            entrytitle plottitle;
      layout overlay /
    xaxisopts=(label=xlabel labelattrs=(color=axislabelcolor size=axislabelsize))
    yaxisopts=(label=ylabel labelattrs=(color=axislabelcolor size=axislabelsize));

    barchart x=var1 y=var2 / fillattrs=(color=barfillcolor) stat=statistics barwidth=0.5;

    endlayout;
        entryfootnote halign=right "Created: " SYSDATE9 /
        textattrs=GraphValueText;
    endgraph;
  end;
run;

%let axislabelsize=15;
%let axislabelcolor=BLUE;
%let barfillcolor=GREEN;
%let statistics=MEAN;
proc sgrender data=consumer_info template=dynagr;
  dynamic var1='area' var2='salary'
  xlabel='Area' ylabel='AVG Salary' plottitle='Area Report';
run;

%let axislabelsize=15;
%let axislabelcolor=BLUE;
%let barfillcolor=RED;
%let statistics=MEDIAN;
proc sgrender data=consumer_info template=dynagr;
     dynamic var1='year' var2='spend'
     xlabel='Year' ylabel='MEDIAN Spend' plottitle='YOY Report';
run;
```